

```
* Stata Analysis Examples Replication for ASDA 2nd Edition
* Berglund April 2017
* Chapter 5
```

```
use "P:\ASDA 2\Data sets\nhanes 2011_2012\nhanes1112_sub_8aug2016.dta", clear
rename *, lower
```

```
* Example 5.1 : generate weighted histogram of cholesterol
generate int_wtmec2yr = int(wtmec2yr)
histogram lbxtc if age18p [fweight = int_wtmec2yr]
```

```
* Example 5.2 : generate boxplot of cholesterol by gender
gen female =1 if riagendr==2
replace female=0 if riagendr==1
tab female
graph box lbxtc [pweight = wtmec2yr] if age18p==1, by(female)
```

```
* Example 5.3 : Population totals using NCSR data
use "P:\ASDA 2\Data sets\ncsr\ncsr_sub_13nov2015.dta", clear
rename *, lower
```

```
gen ncsrwtsh_pop = ncsrwtsh * (209128094 / 9282)
svyset seclustr [pweight = ncsrwtsh_pop], strata(sestrat)
* total MDE
svy: total mde
mat list r(table)
estat effects
* Totals over marital status
svy: total mde, over(mar3cat)
mat list r(table)
estat effects
```

```
* Example 5.4 : Total HH Wealth using HRS 2012 data
use "P:\ASDA 2\Data sets\HRS 2012\hrs_sub_28sep2016.dta", clear
rename *, lower
```

```
svyset secu [pweight = nwgthh], strata(stratum) vce(linearized) singleunit(missing)
gen finr = 1
replace finr = 0 if nfinr != 1
svy, subpop(finr): total hllatota
```

```
* Example 5.5: Estimating the Mean Value of Household Income using the 2012 HRS Data.
svy, subpop(finr): mean hllitot
```

```
* Example 5.6: Estimating Mean Systolic Blood Pressure using the NHANES Data.
use "P:\ASDA 2\Data sets\nhanes 2011_2012\nhanes1112_sub_8aug2016.dta", clear
rename *, lower
svyset sdmvpsu [pweight = wtmec2yr], strata(sdmvstra) vce(linearized) singleunit(missing)
```

```
svy, subpop(age18p): mean bpxsyl
estat effects
```

```
* Example 5.7: Estimating the Mean Value of Total Household Wealth using the HRS Data.
use "P:\ASDA 2\Data sets\HRS 2012\hrs_sub_28sep2016.dta", clear
rename *, lower
svyset secu [pweight=nwgthh], strata(stratum)
gen finr = 1
replace finr = 0 if nfinr != 1
svy, subpop(finr): mean hllatota
estat effects
```

```
* Example 5.8: Estimation of the Population Standard Deviations of NHANES 2011-2012 Measures of
High-density and Total Cholesterol Level.
use "P:\ASDA 2\Data sets\nhanes 2011_2012\nhanes1112_sub_8aug2016.dta", clear
rename *, lower
svyset sdmvpsu [pweight=wtmec2yr], strata(sdmvstra) vce(linearized) singleunit(missing)
sum lbdhdd lbxtc [aweight=wtmec2yr] if age18p==1
```

```

* Example 5.9: Estimating Population Quantiles for Total Household Wealth Using the HRS Data.
* used findit epctile to install package and for download
use "P:\ASDA 2\Data sets\HRS 2012\hrs_sub_28sep2016.dta", clear
rename *, lower

svyset secu [pweight = nwgthh], strata(stratum)
epctile hllatota, percentiles(25 50 75) subpop(if nfinr==1) svy

* Example 5.10: Estimating the Lorenz Curve and Gini Coefficient for the 2012 HRS Population
Distribution of Total Household Wealth.
svyset secu [pweight=nwgthh], strata(stratum)
gen poshllatota=hllatota
replace poshllatota=0 if hllatota<0
gen finr=1
replace finr=0 if nfinr!=1
svylorenz poshllatota, ngp(10) subpop(finr)

* Example 5.11: Estimation of the Correlation of Adult Total and High-Density Cholesterol Measures
in the 2011-2012 NHANES.
use "P:\ASDA 2\Data sets\nhanes 2011_2012\nhanes1112_sub_8aug2016.dta", clear
rename *, lower
sum lbdhdd lbxtc [aweight=wtmec2yr] if age18p==1
correlate lbdhdd lbxtc [aweight=wtmec2yr] if age18p==1
gen stdlbxtc = (lbxtc - 194.4355) / 41.05184
gen stdlbdhdd = (lbdhdd - 52.83826) / 14.93157

summarize stdlbdhdd stdlbxtc [aweight=wtmec2yr] if age18p==1
svyset sdmvpsu [pweight=wtmec2yr], strata(sdmvstra)
svy, subpop(age18p): regress stdlbdhdd stdlbxtc

* Example 5.12: Estimating the Population Ratio of High Density to Total Cholesterol for U.S.
Adults.
svy, subpop(age18p): ratio (lbdhdd/lbxtc)
estat effects

* Example 5.13: Estimating the Proportions of Males and Females Age >= 70 with Diabetes Using the
HRS Data.
use "P:\ASDA 2\Data sets\HRS 2012\hrs_sub_28sep2016.dta", clear
rename *, lower

svyset secu [pweight=nwgtr], strata(stratum) vce(linearized) singleunit(missing)
svy: mean diabetes, subpop(if nage >= 70) over (gender)

* Example 5.14: Estimating Mean Systolic Blood Pressure for Males and Females Age > 45 using the
2011-2012 NHANES data.
use "P:\ASDA 2\Data sets\nhanes 2011_2012\nhanes1112_sub_8aug2016.dta", clear
rename *, lower
svyset sdmvpsu [pweight=wtmec2yr], strata(sdmvstra)
gen gender=riagendr
svy, subpop(if age > 45): mean bpxsyl, over(gender)
estat effects
estat size

* Example 5.15: Estimating Differences in Mean Total Household Wealth Between HRS Subpopulations
Defined by Educational Attainment Level.
use "P:\ASDA 2\Data sets\HRS 2012\hrs_sub_28sep2016.dta", clear
rename *, lower

gen finr = 1
replace finr = 0 if nfinr != 1
svyset secu [pweight = nwgthh], strata(stratum) vce(linearized) singleunit(missing)
svy, subpop(finr): mean hllatota, over(edcat)
lincom [hllatota]1 - [hllatota]4

* Example 5.16: Estimating Differences in Mean Total Household Wealth from 2010 to 2012 using Data
from the HRS study.

```

```
use "P:\ASDA 2\Data sets\HRS 2012\hrs 2010\hrs_2010_2012_both.dta", clear
rename *, lower

* prepare data for analysis
gen hhweight = mwgthh
replace hhweight = nwgthh if year == 2012
gen totwealth = h10atota
replace totwealth = h11atota if year == 2012
gen finr2010 = 0
replace finr2010 = 1 if (year == 2010 & mfinr == 1)
gen finr2012 = 0
replace finr2012 = 1 if (year == 2012 & nfinr == 1)
gen finr2010_2012 = 0
replace finr2010_2012 = 1 if finr2010 == 1 | finr2012 == 1
svyset secu [pweight=hhweight], strata(stratum) vce(linearized) singleunit(missing)
svy, subpop(finr2010_2012): mean totwealth, over(year)
lincom [totwealth]2010 - [totwealth]2012
```